

Evoluční model pesimizace prostředí v Maple

Jiří Kalina

Masarykova univerzita, Institut biostatistiky a analýz,
Kamenice 126/3, Brno, 625 00
kalina@mail.muni.cz

Abstrakt

Príspevek se zabývá popisem a diskuzí populačního modelu se zahrnutím náhodné evoluční složky, založeného na jednoduchém modelu chemostatu popsáném v [2]. Model je rozšířen na neomezený počet populací a je z něj vyňata omezující podmínka rovnosti mortality a podílu nevyužitého substrátu podléhajícího zkáze, což umožňuje jeho aplikaci i na složitější společenstva. Jádrem příspěvku je implementace modelu v prostředí Maple, ve kterém je pomocí numerického řešiče diferenciálních rovnic dsolve spočten vývoj velikosti populací a na výsledných grafech je demonstrována oprávněnost předem vyslovených předpokladů.

Abstract

This paper deals with the description and discussion of the population model including a random evolutionary component, based on a simple model of chemostat described in [2]. The model is extended to an unlimited number of populations and a condition of equality of mortality and the proportion of unused perishable substrate, which allows its application to more complex systems. The core contribution is an implementation of the model in a Maple environment with an use of a numerical differential equation solver "dsolve". The evolution of population sizes is counted and plots of results demonstrate the legitimacy of former assumptions.

Klíčová slova

Populační model, pesimizace.

Keywords

Population model, pesimization.

1 Základní vlastnosti modelu

Předpokládejme uzavřený systém složený z prostředí a n navzájem přímo neinteragujících populací, které ho obývají, $n \in N_0$. Jedince náležející k jedné populaci budeme souhrnně nazývat druhem. Úživnost prostředí, tj. veličinu, která charakterizuje podmínky pro přežití a reprodukci jedinců, označme S a považujme ji zjednodušeně za množství potravy (substrátu, resp. energie), které je k dispozici společně pro všechny populace v prostředí. S budeme považovat za funkci času $S(t)$. Přírůstek substrátu v prostředí za jednotku času (růst, resp. transport energie do systému) považujme za konstantní a označme jej d .

Vzhledem k předpokladu, že populace v modelu spolu přímo neinteragují, omezuje se jejich vzájemné ovlivnění na soupeření o dostupné množství potravy, jde tedy o ryze kompetitivní vztah [1]. Vzhledem k jednoduchosti modelu budeme druhy odlišovat pouze schopností jejich zástupců spotřebovávat společnou potravu a takto přijatý substrát využít k vlastnímu přežití a reprodukci. Tento proces popíšeme dvojicí veličin q a e , kde q_i představuje množství substrátu přijaté jedním jedincem i -té populace za jednotku času. Tuto veličinu budeme dále považovat za funkci množství dostupného substrátu $q_i(S)$. Naproti tomu hodnota e_i označuje efektivitu přeměny substrátu na živé jedince i -tého druhu jako poměr spotřebovaného substrátu vůči počtu nově narozených jedinců. V první variantě

modelu budeme hodnoty q_i a e_i považovat za vzájemně nezávislé. Dále budeme předpokládat konstantní úmrtnost m , stejnou pro všechny populace, jako podíl jedinců daného druhu uhynulých za jednotku času a nakonec zbytkový podíl nespotřebovaného substrátu z , který podlehe zkáze, aniž by byl spotřebován některou z uvažovaných populací.

Na počátku modelovaného období budeme předpokládat jednu (nebo větší konstantní počet) populaci v prostředí, která spotřebovává dostupný substrát pro vlastní reprodukci. Snadno lze ukázat, že po určité době dojde k ustálení počtu jedinců populace (uvědomme si, že jde o silně zjednodušený idealizovaný model) na hodnotě odpovídající úživnosti prostředí, která se v dalším průběhu času již nebude měnit. Je nutno poznamenat, že pro jednoduchost model opomíjí veškeré komplikace, které s sebou přináší rozmanitost pohlavního rozmnožování jako je samo pohlaví, hledání partnera či nestejný počet potomků každého jedince. Budeme tedy nadále předpokládat, že dostatečným požadavkem pro další rozmnožování populace je přítomnost alespoň jednoho jedince daného druhu v systému, že potomkem jedince určitého druhu je (až na výjimky) jedinec téhož druhu a že křížení mezi jedinci různých druhů není možné.

Situace se nicméně začne komplikovat ve chvíli, kdy se v prostředí objeví zástupce nového druhu. V modelovém uzavřeném systému není možný průnik takového jedince zvenčí, proto jej budeme považovat za mutanta některé z již existujících populací, který se bude odlišovat parametry q_i a e_i , poruší tedy předpoklad z předcházejícího odstavce. Výskyt mutantů bude relativně řídký a bude se odehrávat pouze v určitých okamžicích. Takový jedinec se v našem modelu může, bude-li úspěšný ve vzájemné kompetici, stát zakladatelem populace nového druhu, v opačném případě může ovšem nová populace záhy vyhynout a uvolnit tak prostředky pro zástupce jiných druhů.

Časovou změnu množství substrátu dostupného v systému je na základě uvedených předpokladů možné vyjádřit následujícím vztahem:

$$\frac{\partial S(t)}{\partial t} = d - z \cdot S(t) - \sum_{i=1}^n q_i(S) \cdot P_i \quad (1)$$

kde $S(t)$ je dostupné množství substrátu v čase t , d je přírůstek substrátu za jednotku času, z je podíl nevyužitého substrátu za jednotku času, n je počet populací přítomných v systému, $q_i(S)$ je spotřeba substrátu pro jednoho jedince i -tého druhu za jednotku času a P_i je počet jedinců i -té populace.

Popišme nyní tvar závislosti $q_i(S)$ množství substrátu spotřebovaného jedincem i -té populace za jednotku času. Opět zjednodušeně předpokládejme, že dostupnost substrátu v prostředí je stejná pro všechny jedince všech populací a její vyhledávání proto nehraje v modelu žádnou roli. V případě, že bude množství dostupného substrátu nulové, $S = 0$, bude přirozeně rovněž $q_i = 0$. Jakmile začne v prostředí narůstat množství dostupné potravy, poroste přímo úměrně také množství zkonsumované jedinci všech populací. Vzhledem k fyziologickým možnostem organismů je však maximální množství potravy přijaté za jednotku času shora omezené, proto se bude s narůstajícím množstvím substrátu v prostředí křivka funkce q_i oddělovat od přímky přímé úměry, až plynule konverguje ke konstantní funkci. Uvedený předpoklad dokonale splňuje v biologii a příbuzných oborech hojně využívaná logistická křivka následujícího tvaru:

$$q_i(S) = \frac{a \cdot S}{1 + B_i \cdot a \cdot S} \quad (2)$$

kde parametr a určuje strmost počáteční části křivky a rovněž rychlost dosažení následující asymptotické fáze, zatímco parametr B_i její výšku. V první, jednodušší, variantě modelu budeme předpokládat jednak, že B_i je totožné pro všechny populace (což není sám o sobě předpoklad, který by výrazně ovlivnil chování modelu), současně ale také, že B_i je konstantní a nezávislé na ostatních parametrech modelu. Rovněž parametr a budeme v první variantě modelu považovat za konstantu.

Přejdeme nyní k rovnici popisující vývoj počtu jedinců i -té populace v čase. Změna počtu jedinců, P_i bude vyjádřena jako množství substrátu zkonsumovaného jedinci populace, vynásobeného efektivitou přeměny substrátu na živé jedince, minus počet uhynulých jedinců:

$$\frac{\partial P_i(t)}{\partial t} = q_i(S) \cdot e_i \cdot P_i(t) - m \cdot P_i(t) \quad (3)$$

kde $P_i(t)$ je počet jedinců populace v čase t , $q_i(S)$ je spotřeba substrátu pro jednoho jedince i -tého druhu za jednotku času, e_i je efektivita přeměny substrátu na živou hmotu i -tou populací a m je relativní úmrtnost, pro jednoduchost opět totožná pro všechny populace.

Uvedený postup nám poskytuje soustavu $n + 1$ diferenciálních rovnic, jejímž řešením lze získat informaci o vývoji počtu jedinců jednotlivých populací v čase. Vzhledem k tomu, že v první variantě modelu uvažujeme q_i totožné pro všechny populace v systému, je řešení závislé především na hodnotách e_i pro jednotlivé populace. V případě, že se v systému nevyskytne mutant, lze ukázat, že v dlouhodobém horizontu se v závislosti na hodnotách konstantních parametrů d, z, a a b_i a na počátečních podmínkách $S(0)$ a $P_i(0)$ v systému ustálí stav, ve kterém je množství dostupného substrátu v prostředí konstantní a navíc počet jedinců nejvýše jedné populace nekonverguje k nule.

Platí totiž:

$$\frac{d\left(S + \sum_{i=1}^n \frac{P_i}{q_i}\right)}{dt} = d - m\left(S + \sum_{i=1}^n \frac{P_i}{q_i}\right) - (m - z) \sum_{i=1}^n \frac{P_i}{q_i} \quad (4)$$

a tedy po integraci podle t potom:

$$S + \sum_{i=1}^n \frac{P_i}{q_i} = \int e^{zt} d dt \cdot e^{-zt} + \int e^{zt} (z - m) \sum_{i=1}^n \frac{P_i}{q_i} dt \cdot e^{-zt} + C \cdot e^{-zt} \quad (5)$$

a po úpravě:

$$S + \sum_{i=1}^n \frac{P_i}{q_i} = \frac{d}{z} + (z - m) \int e^{zt} \sum_{i=1}^n \frac{P_i}{q_i} dt \cdot e^{-zt} + C \cdot e^{-zt} \quad (6)$$

za předpokladu, že pro $t \rightarrow \infty$ konvergují 2. a 3. sčítanec k nule, dostáváme výsledek:

$$\lim_{t \rightarrow \infty} \left(S + \sum_{i=1}^n \frac{P_i}{q_i} \right) = \frac{d}{z} \quad (7)$$

Tedy, pokud bude úživnost prostředí dostatečná, zvítězí v kompetici o omezený zdroj potravy v uzavřeném systému pouze jediný druh (viz (8) a (9)), zatímco ostatní postupně vyhynou. Lze ukázat (viz [2]), že půjde o druh s nejvyšší hodnotou e_i , tj. ten, který je schopen nejefektivněji zpracovávat stravu ve prospěch přírůstku vlastních jedinců.

Zkomplikujme nyní situaci možností, že se v určitém čase v systému objeví mutant, tj. jedinec, jehož e_i se bude lišit od e_i jeho rodiče a půjde tak vlastně podle našich kritérií o zakladatele nového druhu. V modelu předpokládáme dobu mezi výskytem dvou mutantů náhodnou s rovnoměrným rozdělením pravděpodobnosti, což sice neodpovídá reálnému prostředí, nicméně zřejmě samotná perioda výskytu mutantů nemá na stav systému z dlouhodobého hlediska vliv.

Hodnotu e_i mutovaného potomka pak budeme uvažovat v intervalu $\left(\frac{e_j}{2}, \frac{3 \cdot e_j}{2}\right)$, kde e_j odpovídá hodnotě efektivity přeměny substrátu na živou hmotu jeho rodiče. Rozdělení pravděpodobnosti využijeme pro jednoduchost opět rovnoměrné. V modelu budeme uvažovat stejnou pravděpodobnost zplození mutantu pro všechny žijící jedince, tj. pravděpodobnost vzniku mutantu i -té populace bude přímo úměrná velikosti této populace.

2 Implementace modelu v prostředí Maple

Základem modelu v softwarovém prostředí Maple 16 je cyklus, jehož jedna iterace odpovídá období mezi výskytem dvou mutantů. Časová délka období reprezentovaného iterací je proměnlivá. Jednotlivé populace jsou v modelu reprezentovány řádky populační matice *Matice*, jejíž první sloupec udává pro každou populaci efektivitu e_i , druhý sloupec pak odpovídá počtu jedinců dané populace.

```
celkemcyklu := 250;

a := 2.5;
b := 0.8;
d := 12;
m := 0.1;
z := 0.1;
cas := 0

q := proc (x) options operator, arrow; a*x/(1+a*b*x) end proc
```

Před spuštěním cyklu je vygenerována dvouřádková matice se dvěma náhodnými populacemi, které vstupují do prvního období. Délka období je volena náhodně z uživatelem zadaného intervalu. Na začátku každého období je provedena kontrola všech řádků matice a „neúspěšné“ populace s počtem jedinců nižším než 1 (tj. populace nesplňující zadanou podmínku pro další rozmnožování)¹ jsou z modelu odstraněny. Následně probíhá vygenerování náhodného mutantu, který je zapsán jako nový řádek do matice (tj. populace s jediným jedincem).

```
# Matice se dvema nahodnymi populacemi;
h1 := rand(1 .. 10);
h2 := (1/100)*rand(1 .. 100);
Matice := Matrix(2, 2, [[h2(), h1()], [h2(), h1()]]);

# Hlavni cyklus;
for cyklus from 1 to celkemcyklu do

  # Cyklus pro odstraneni populaci s mene nez 1 jedincem;
  v:=0:
  for i from 1 by 1 to RowDimension(Matice) do
    if (Matice[i-v,2]<1) then
      Matice:=DeleteRow(Matice,i-v);
      v:=v+1
    end if:
  end do:

  # Cyklus pro vznik jednoho jedince s nahodnou mutaci;
  jedincu:=add(Matice[k,2],k=1..RowDimension(Matice));
  h:=(rand(1..10000* floor(jedincu)))/(10000):
  budemutant:=h():
  soucet:=0:
  j:=0:
  for j from 1 by 1 while soucet<budemutant do
    soucet:=soucet+Matice[j,2]:
  end do:
```

¹ Pro snadnější průběh výpočtů budeme počet jedinců populace považovat za spojitou veličinu a budeme tedy pracovat i s neceločíselnými hodnotami. Pro názornější představu lze hodnotu P_i považovat např. za celkovou hmotnost všech jedinců populace, což odpovídá rovněž procesu přeměny potravy na živou hmotu (tj. rodičů i potomků současně).

```

mutant:=Matice[j-1,1]:

# Mutace a zapis mutanta do matice:
h:=(rand(0..100)-50)/(100):
mutant:=mutant+mutant*h():
Matice(RowDimension(Matice)+1,1..):=<mutant,1>:

```

V dalším kroku je pomocí vnořeného cyklu vytvořena soustava $n + 1$ diferenciálních rovnic postupně pro jednotlivé populace z matice a navíc pro celkové dostupné množství substrátu v systému, z velikosti populací a proměnné Y , udávající množství dostupného substrátu po skončení předešlého období jsou vygenerovány počáteční podmínky pro řešení a následně je soustava řešena pomocí numerické varianty příkazu *dsolve*. Řešení je z důvodu omezení časové náročnosti hledáno pouze pro období mezi výskyty mutantů, po jehož uplynutí se změní počáteční podmínky.

```

# Vytvoreni soustavy diferencialnich rovnic;
rovnice[0]:=diff(S(t),t)=d-z*S(t)-
q(S(t))*add(P[i](t),i=1..RowDimension(Matice)):
for i from 1 by 1 to RowDimension(Matice) do
  rovnice[i]:=diff(P[i](t),t)=-m*P[i](t)+q(S(t))*P[i](t)*Matice[i,1]:
end do:
soustava:=seq(rovnice[i],i=0..RowDimension(Matice)):

# Vytvoreni pocatecnich podminek;
if cas=0 then
  podminka[0]:=S(0)=1: else podminka[0]:=S(cas)=Y:
end if:
for i from 1 by 1 to RowDimension(Matice) do
  podminka[i]:=P[i](cas)=Matice[i,2]:
end do:
podminky:=seq(podminka[i],i=0..RowDimension(Matice)):

# Cas do pristi mutace;
minule:=cas:
h:=rand(1..40):
cas:=cas+h(1..40):

# Reseni a vykresleni do grafu;
reseni:=dsolve({soustava,podminky},numeric, maxfun=10000000):
krivky:=seq([t,P[i](t)],i=1..RowDimension(Matice)):
graf[cyklus]:=odeplot(reseni,([krivky],minule..cas)):
assign(reseni(cas)):

# Zapis novych hodnot do matice;
for i from 1 by 1 to RowDimension(Matice) do
  Matice[i,2]:=P[i](t):
  unassign('P[i](t)'):
end do:
Y:=S(t):
unassign('t'):
unassign('P'):
unassign('S'):

end do:

```

Vzhledem k tomu, že S konverguje v období mezi dvěma mutacemi (jak bylo výše ukázáno) ke konstantě, konverguje podle definice ke konstantní hodnotě rovněž funkce příjmu potravy q a tedy můžeme provést následující výpočet:

$$\begin{aligned} \frac{dP_i}{dt} &\xrightarrow{t \rightarrow \infty} P_i \cdot e_i \cdot q - mP_i \\ P_i &\xrightarrow{t \rightarrow \infty} e^{P_i(qe_i - m)} \end{aligned} \quad (8)$$

Tedy zřejmě velikost populace v rámci období mezi mutacemi konverguje k monotónnímu chování. Předpokládejme nyní:


$$\begin{aligned} P_i \xrightarrow{t \rightarrow \infty} konst. &\Rightarrow P_i \cdot e_i \cdot q - mP_i \xrightarrow{t \rightarrow \infty} 0 \\ P_i \xrightarrow{t \rightarrow \infty} 0 \vee e_i &\xrightarrow{t \rightarrow \infty} \frac{m}{q_i} \end{aligned} \quad (9)$$

Protože pro libovolné populace $i, j: i \neq j$ platí $e_i \neq e_j$, konverguje v období mezi dvěma mutacemi velikost všech populací, případně vyjma populace s nejvyšším e_i , k nule.

Příkazem `display(seq(graf[i], i = 1 .. celkemcyklu))` si nyní můžeme zobrazit přes sebe graf vývoje populací ve všech obdobích (obr. 1) a na první pohled je zřejmé, že skutečně v každém období mezi dvěma mutacemi je jediná populace, jejíž velikost roste „na úkor“ ostatních populací. Ve variantě bez mutací by se po určité době velikost této populace ustálila na hodnotě úživnosti prostředí, zatímco velikost konkurenčních populací by klesla k nule. Náhodně se vyskytující mutace však do systému vnáší dynamické chování, neboť v případě, kdy se objeví mutant s e_i vyšším, než bylo dosavadní nejvyšší e_j , rozšíří se tato i -tá populace a postupně zapříčiní vyhynutí všech populací včetně j -té, pokud není ještě dříve vystřídána jinou populací, vzešlou z mutanta s vyšším e_k .

Vzhledem k tomu, že v této variantě modelu není výše efektivity přeměny potravy na živou hmotu e ničím omezena, dochází k tomu, že se neustále objevují mutanti s vyšší hodnotou e_i , která může v rozporu s realitou překročit dokonce hodnotu 1 a růst neomezeně do nekonečna. Velikost momentální největší populace se tak bude v modelu pohybovat v čase po exponenciální křivce donekonečna. Takový proces je samozřejmě zcela nerealistický a proto je třeba model upravit.

Zvyšování efektivity přeměny potravy na živou hmotu vlastního druhu se v modelu ukazuje jako správná strategie vedoucí k přežití. Je ovšem patrné, že s rostoucím e_i vítězná populace dochází k poklesu substrátu v prostředí, neboť vítězná populace ostatní jednoduše „vyhladoví“. V našem nerealistickém případě z dlouhodobého hlediska dochází až ke konvergenci S k nule. Tento pokles úživnosti a obecně evoluční zhoršení podmínek prostředí na ty nejhorší možné se nazývá pesimizací prostředí.



Obr. 1: Exponenciální růst populací v modelu ve variantě bez omezení efektivity e_i přeměny substrátu na živou hmotu.

3 Varianta s omezením efektivity

Fyzikální podstata živých tvorů neumožňuje růst efektivit e_i nad hodnotu 1 (ve skutečnosti je číslo vzhledem k chemickým procesům v organismu samozřejmě ještě nižší), vlastní fyziologie organismů však znamená ještě výraznější pokles efektivit, neboť v reálném prostředí nemůže věnovat žádný jedinec maximum prostředků na hospodárné nakládání se získanou energií (tj. vlastní růst a reprodukci), ale musí rovněž zajistit řadu dalších procesů, mezi které kromě jiných strategických činností patří samotný příjem potravy.

V předchozí zjednodušené variantě modelu jsme předpokládali závislost množství přijaté potravy q pouze na množství substrátu S v prostředí, zbývající parametry a a B_i jsme považovali za konstantní pro všechny populace. Definujme nyní vztah mezi efektivitou přeměny substrátu na živou hmotu e_i a množstvím potravy, kterou je jedinec schopen přijmout za jednotku času q_i , který bude reprezentovat strategii druhu ve smyslu rozhodování mezi investicí energie do většího množství získané potravy nebo lepším využitím potravy pro vlastní růst a reprodukci.

Nahradíme tedy v rovnici logistické křivky (2) konstantu B_i kvadratickou funkcí e_i takto:

$$B_i = e_i^2 + c \quad (10)$$

tj. s rostoucí efektivitou $e_i > 0$ bude injektivně klesat množství substrátu q_i , které je jedinec daného druhu schopen zkonsumovat. Tato vazba zamezí nekontrolovanému růstu e_i a přiblíží model realitě. Optimální strategií pak bude dosáhnout takového e_i , aby byla hodnota $e_i \cdot q_i$ (viz (3)) maximální, tj:

$$q_i \cdot e_i = \frac{e_i \cdot a \cdot S}{1 + (e_i^2 + c) \cdot a \cdot S} \quad (11)$$

a po derivaci podle e_i :

$$\frac{a \cdot S \cdot (1 - a \cdot S \cdot e^2 + a \cdot S \cdot c)}{(1 + a \cdot S \cdot e^2 + a \cdot S \cdot c)^2} \quad (12)$$

Po úpravě, protože jmenovatel je nenulový, získáváme vztah pro optimální \hat{e} :

$$\hat{e} = \sqrt{\frac{1}{a \cdot S} + c} \quad (13)$$


Ve zdrojovém kódu Maple stačí změnit definici funkce q_i na funkci dvou proměnných S a e_i a upravit cyklus určující tvar diferenciálních rovnic:

```
q := proc (x, y) options operator, arrow; a*x/(1+a*(y^2+c)*x) end proc
# Vytvoreni soustavy diferencialnich rovnic;
rovnice[0] := diff(S(t), t) = d-z*S(t)-add(q(S(t), Matice[i, 1])*P[i](t), i
= 1 .. RowDimension(Matice));
for i to RowDimension(Matice) do
  rovnice[i] := diff(P[i](t), t) = -m*P[i](t)+q(S(t), Matice[i,
1])*P[i](t)*Matice[i, 1]:
end do;

soustava := seq(rovnice[i], i = 0 .. RowDimension(Matice))
```

Podle očekávání je po vykreslení grafu jasně patrný nárůst celkového e_i momentálně nejméně úspěšných populací, který však zpomaluje až ke konstantní hodnotě odpovídající úživnosti prostředí pro populaci s hodnotou blízkou \hat{e} podle rovnice (13). Z dlouhodobého hlediska by za neměnných podmínek v systému opravu po řadě mutací převládla jediná populace s $e_i = \hat{e}$.

Protože \hat{e} je nejvyšší ze všech dlouhodobě možných efektivit přeměny substrátu na živou hmotu, dojde postupem času v prostředí opět k poklesu dostupného množství potravy na nejnižší možnou mez – vzájemný souboj populací o zdroje tedy stejně jako v předchozím případě nastoluje ty nejhroší podmínky pro přežití, tj. dochází k evoluční pesimizaci.



Obr. 2: Omezený růst populací v modelu s omezením efektivity e_i přeměny substrátu na živou hmotu. Po dosažení hodnoty \hat{e} se růst zastavuje na hodnotě úživnosti prostředí.

4 Závěr

Ve variantě modelu, kdy není efektivita přeměny substrátu na živou hmotu e shora omezena, dochází vlivem prostředí za vzniku náhodných mutací k postupnému zvyšování e přibližně exponenciálním tempem, přičemž každá populace po určité době vyhyne, protože úživnost prostředí bude snížena populací s vyšší efektivitou e pod přípustnou mez.

Pokud je do modelu vložena zpětná vazba omezující růst efektivity e jeho klesající úměrností vůči množství zkonsumovaného substrátu q , dochází postupnými mutacemi k přibližování maximální efektivity e k mezní hodnotě úživnosti prostředí \hat{e} , která je daná fyziologickými vlastnostmi organismů. V tomto případě platí odvozené vztahy dokazující, že nejvýše jedna populace v dlouhodobém měřítku nevyhyne a obsadí celý systém. Pro tuto populaci platí podmínka $e = \hat{e}$, tj. není ohrožena žádným mutantem.

Vlivem evolučních změn vedoucích k postupnému hynutí populací s nižší hodnotou e než mají jejich konkurenční mutanti, dochází k tzv. pesimizaci prostředí, tj. snížení úživnosti na nejmenší přístupnou mez. V krajním případě může tento jev vést až k úplnému vyhynutí všech populací.

5 Literatura

- [1] Hřebíček, J., Pospíšil, Z., Urbánek J.: Úvod do matematického modelování s využitím Maple. Akademické nakladatelství CERM, s.r.o., Brno, 2010.
- [2] Diekmann, O. A beginner's guide to adaptive dynamics. Mathematical modelling of population dynamics, Banach Center Publications, vol. 63. Institute of Mathematics Polish Academy of Sciences, Warszawa, 2004.